// SPDX-License-Identifier: MIT

/**
    ███████╗██████╗░███████╗███╗░░██╗░██████╗░██╗░░░░░░░██╗░█████╗░██████╗░
    ██╔════╝██╔══██╗██╔════╝████╗░██║██╔════╝░██║░░██╗░░██║██╔══██╗██╔══██╗
    █████╗░░██████╔╝█████╗░░██╔██╗██║╚█████╗░░╚██╗████╗██╔╝███████║██████╔╝
    ██╔══╝░░██╔══██╗██╔══╝░░██║╚████║░╚═══██╗░░████╔═████║░██╔══██║██╔═══╝░
    ██║░░░░░██║░░██║███████╗██║░╚███║██████╔╝░░╚██╔╝░╚██╔╝░██║░░██║██║░░░░░
    ╚═╝░░░░░╚═╝░░╚═╝╚══════╝╚═╝░░╚══╝╚═════╝░░░░╚═╝░░░╚═╝░░╚═╝░░╚═╝╚═╝░░░░░
@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@@@@@@@@/                        &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@@@@                             &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@         &@@@@@@@@@@@@@@@@@     &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@       @@@@@@@@@@@@@@@@@@@@@@     &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@      @@@@@@@&                     &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@      @@@@@&                        &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@.     (@@@@@@@@@@@@@@@@@@@@@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@     @@@@@@@@@@@@@@@@@@@@@@@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@&                   (@@@@@@@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@@@                      (@@@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@@@@@@@@%###########       @@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@@@@@@@@@@@@@@@@@@@@@@     &@@@@&     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@@@@@@@@@@@@@@@@@@@@/      @@@@@.     @@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@                         @@@@@&     @@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@                     &@@@@@@@      @@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@     @@@@@@@@@@@@@@@@@@@@@@       @@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@     @@@@@@@@@@@@@@@@@#         @@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@                             @@@@@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@     @@@@@                        %@@@@@@@@@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@

https://frenswap.com
wan swap som smolting fren? Need soup today?
*/

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}



pragma solidity ^0.8.0;

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

pragma solidity ^0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}


pragma solidity ^0.8.0;

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin guidelines: functions revert instead
 * of returning `false` on failure. This behavior is nonetheless conventional
 * and does not conflict with the expectations of ERC20 applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5,05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);

        uint256 currentAllowance = _allowances[sender][_msgSender()];
        require(currentAllowance >= amount, "ERC20: transfer amount exceeds allowance");
        unchecked {
            _approve(sender, _msgSender(), currentAllowance - amount);
        }

        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender] + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        uint256 currentAllowance = _allowances[_msgSender()][spender];
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(_msgSender(), spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves tokens `amount` from `sender` to `recipient`.
     *
     * This is internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(
        address sender,
        address recipient,
        uint256 amount
    ) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        uint256 senderBalance = _balances[sender];
        require(senderBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[sender] = senderBalance - amount;
        }
        _balances[recipient] += amount;

        emit Transfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be to transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

/// @title frenswap

pragma solidity ^0.8.0;

/**
* Reviewing the code to see what it does? Yes, good for you! The above is all standard
* library routines. The stuff you want to scrutinise is all below. But do look at the 
* libary stuff too, and if you have questions reach out. The more you know about this 
* the safer crypto is for everyone! I have tried to write clear code, which in some cases
* means more lines for operations that could be smooshed into a single declaration. This
* takes more space but is more humanly readable, and the compiler certainly doesn't care
* either way!
*/

contract frenswap is Ownable {

    // Contract location for the fren token:
    ERC20 public frenContract = ERC20(0x3cceEDCa53d09000108573982A6057A511675861); 
    
    // Address of the fren vault
    address public frenVault = 0xde2fbC999571704De90Cb1593E80EcBD6950295A;

    // Interval settings (will also be the initial value when resetting on swap)
    uint256 public intervalAmount = 1000000000000000;
    uint256 public maxBid = 100000000000000000;
    uint256 public lastResetAt = 0;

    // Events - these are used to show something has happened, and can be watched on chain:
    event brandNewFren(address newFren, uint256 amount, uint256 swappedAt);
    event frenVaultUpdated(address);
    event intervalAmountUpdated(uint256);
    event maxBidUpdated(uint256);
    event lastResetAtInitialised(uint256);
    event ethWithdrawn(uint256);
    
    receive() external payable {}

    /**
     * The fallback function is executed on a call to the contract if
     * none of the other functions match the given function signature.
     */
    fallback() external payable {}

    /**
    * The routines here all allow the setting of state variables defined above
    * i.e. they are the configuration routines.
    */
    function setVaultAddress(address _frenVault) public onlyOwner returns (bool) {
        frenVault = _frenVault;
        emit frenVaultUpdated(_frenVault);
        return true;
    }

    function setIntervalAmount(uint256 _intervalAmount) public onlyOwner returns (bool) {
        intervalAmount = _intervalAmount;
        emit intervalAmountUpdated(_intervalAmount);
        return true;
    }

    function setMaxBid(uint256 _maxBid) public onlyOwner returns (bool) {
        maxBid = _maxBid;
        emit maxBidUpdated(_maxBid);
        return true;
    }

    function initialiseLastResetAt() public onlyOwner returns (bool) {
        lastResetAt = block.timestamp;
        emit lastResetAtInitialised(lastResetAt);
        return true;
    }

    function withdraw(uint256 _withdrawal) public onlyOwner returns (bool) {
        payable(msg.sender).transfer(_withdrawal);
        emit ethWithdrawn(_withdrawal);
        return true;
    }

    function getCurrentBid() public view returns(uint256) {
        /**
        * Determine the current bid as the difference in HOURS between the 
        * current block timestamp (a) and the lastResetAt block timestamp (b), 
        * multiplied by the intervalAmount (c):
        * Bid = ((a - b) * c)
        */
        require(lastResetAt <= block.timestamp);
        uint256 hoursSinceBidAccepted = 1 + ((block.timestamp - lastResetAt) / (60 * 60));
        uint256 calculatedBidAmount = (intervalAmount * hoursSinceBidAccepted);
        uint256 currentBidAmount;
        uint256 maximumPossibleBid;
        // See if we have enough ETH on the contract to cover our max bid:
        if (address(this).balance < maxBid) 
        {
            maximumPossibleBid = address(this).balance;
        }
        else
        {
            maximumPossibleBid = maxBid;
        }
        /**
        * See if we set this to the calculated bid or the max bid.
        * (the intention is always to have this as the calculated bid but we 
        * have to allow for the possibility of the contract being low on eth, 
        * and never present a bid we can't fulfil.)
        */
        if (calculatedBidAmount > maximumPossibleBid)
        {
            currentBidAmount = maximumPossibleBid;
        }
        else
        {
            currentBidAmount = calculatedBidAmount;
        }
        return(currentBidAmount);
    }

    function swapFren(uint256 _displayedBidAmount) public {
        /**
        * Requirements:
        * - Caller cannot be the zero address
        * - Caller's fren must be greater than 0
        * - The Bid amount must be equal to our greater than what the user has on the screen, so 
        * - we know that they are getting the ETH they expect (or more).
        */
        require(msg.sender != address(0), "Caller cannot be zero address, sorry blackhole fren, no soup today");
        require(frenContract.balanceOf(address(msg.sender)) >= 1*(10 ** 18), "Holder does not have fren token :(");
        uint256 bidAmount = getCurrentBid();
        // Protect the user from another bid having been accepted since the page loaded:
        require((_displayedBidAmount <= bidAmount), "Sorry fren, another bid already accepted, please refresh the page"); 
        // Reset the bid parameter:
        lastResetAt = block.timestamp;
        // Transfer the fren token to the vault:
        frenContract.transferFrom(msg.sender, address(frenVault), 1*(10 ** 18));
        // Pay the ETH to our new fren:
        payable(msg.sender).transfer(bidAmount);
        // Let the world know:
        emit brandNewFren(msg.sender, bidAmount, block.timestamp);
    }

}